Calculus Single Variable
Brian E. Blank and Steven G. Krantz

Section 6.5
Applications of the Exponential Function

Part I

Load

Each execution group of this section must be entered before continuing with the worksheet.

```
> with(plots):  #Ignore any warnings!
Warning, the name changecoords has been redefined
```

```
> doublingTime := proc()
  local solnSet, tau, f, t, eqn:
  f := args[1]:
  t := args[2]:
  eqn := f(t+tau)=2*f(t):
  solnSet := {solve(eqn, tau)}:
  if type(t,numeric) then
     return fsolve(eqn, tau, 0..infinity);
  elif nops(solnSet) = 1 then
    return op(1,solnSet);
  else
    printf("Doubling time not found.\n");
  end if:
  end proc:
```

1. A Basic Differential Equation

If A is any constant (usually positive in applications) and if k is any nonzero constant, then the unique solution of the initial value problem

$$\frac{d}{dt}\mathbf{y}(t) = k\,\mathbf{y}(t) \quad , \quad \mathbf{y}(0) = A$$

is

$$\mathbf{y}(t) = A\,\mathbf{e}^{(k\,t)}.$$

We find this solution by the Method of Separation of Variables:

```
> Eqn1 := Int(1/y,y) = Int(k,t) + C;
```

$$Eqn1 := \int \frac{1}{y}\,dy = \int k\,dt + C$$

```
> eqn1 := map(value, Eqn1);
```

$$eqn1 := \ln(y) = k\,t + C$$

```
> eqn2 := y = solve(eqn1, y);
```

$$eqn2 := y = \mathbf{e}^{(k\,t + C)}$$

```
> eqn3 := subs( {y=A, t=0}, eqn2);
```

$$eqn3 := A = \mathbf{e}^{C}$$

```
> eqn4 := C = solve(eqn3, C);
```

$$eqn4 := C = \ln(A)$$

```
> eqn5 := subs(eqn4, eqn2);
```

$$eqn5 := y = \mathbf{e}^{(k\,t + \ln(A))}$$

```
> eqn6 := expand(eqn5);
```

$$eqn6 := y = \mathbf{e}^{(k\,t)}\,A$$

Or, we might have asked Maple to find the solution directly by executing this code:

```
> IVP := {diff(y(t),t) = k*y(t),y(0)=A};
  dsolve( IVP , y(t) );
```

When $0 < k$, the solution of our initial value problem is said to ***grow exponentially***.
When $k < 0$, the solution of our initial value problem is said to ***decay exponentially***.

2. Doubling Time for Exponential Growth

If a positive function f is increasing, then we may call the function $t \to \tau(t)$ the ***doubling time*** of f if

$$\mathbf{f}(t + \tau(t)) = 2\,\mathbf{f}(t).$$

In general we expect the doubling time to be nonconstant, as in Figure 1. In this figure we have

$$y_2 = 2\,y_1, \quad y_3 = 2\,y_2, \quad y_4 = 2\,y_3, \quad y_5 = 2\,y_4.$$

Notice that the doubling times, $t_2 - t_1$, $t_3 - t_2$, $t_4 - t_3$, $t_5 - t_4$ are nonconstant.



Figure1: Increasing Doubling Time

Example: Doubling time of a linear function

```
> L := x -> m*x+b;
```

$$L := x \to m\,x + b$$

```
> doublingTime(L,t);  #Note the dependence on  t
```

$$\frac{m\,t + b}{m}$$

Example: Doubling time of the logarithm

```
> doublingTime(ln,t);  #Note the dependence on  t
```

$$-t + t^2$$

```
> plot(doublingTime(ln,t), t= 1 ..2);
  #ln(t) is everywhere increasing but it is positive only for t >
  1
```



Example: Doubling time of a quadratic polynomial

```
> doublingTimes := seq(doublingTime(t->t^2,k*5/100), k=1..100):
  plot([seq([k*5/100,doublingTimes[k]],k=1..100)]);
  #Note non-horizontal graph
```



*Unlike the preceding examples, the doubling time function for exponential growth is **constant**.*

Theorem : **Suppose that k is a positive constant. The doubling time τ of the exponential function $t \to A\, e^{(k\,t)}$ is constant. More precisely, for every $0 < t$, the doubling time is given by**

$$\tau = \frac{1}{k} \ln(2)\ .$$

Proof:

```
> f := t -> A*exp(k*t);
```

$$f := t \to A\, \mathbf{e}^{(k\,t)}$$

```
> tau = solve( f(t+tau) = 2*f(t), tau); #From the solution, note that tau is independent of t
```

$$\tau = \frac{\ln(2)}{k}$$

Theorem : **Suppose that k is a positive constant. Then the exponential function $t \to A\, e^{(k\,t)}$ can be written as $t \to A\, 2^{\left(\frac{t}{\tau}\right)}$ where $\tau = \frac{1}{k} \ln(2)$ is the doubling time.**

Proof:

```
> formula1 := A*exp(k*t);
  formula2 := A*2^(t/tau);
```

$$formula1 := \mathbf{e}^{(k\,t)} A$$

$$formula2 := A\, 2^{\left(\frac{t}{\tau}\right)}$$

```
> simplify( subs(tau = ln(2)/k, formula1 - formula2));
```

$$0$$

3. Exercises Involving Exponential Growth

Exercise

(Exercise 2, page 471) Assuming exponential growth, find the number of bacteria present at 8:00 P.M. given that there are 4000 bacteria at 5:00 P.M. and the population doubles every 2 hours.

Solution

Let $t = 0$ correspond to 5:00 P.M. Let $y(t)$ be the number of bacteria at time t.

```
> A := 4000;  tau := 2;
```

$$A := 4000$$

$$\tau := 2$$

```
> eqn1 := y(t) = A*exp(k*t);
```

$$eqn1 := y(t) = 4000\, e^{(k\, t)}$$

```
> eqn2 := k = ln(2)/tau;
```

$$eqn2 := k = \frac{1}{2} \ln(2)$$

```
> eqn3 := subs(eqn2, eqn1);
```

$$eqn3 := y(t) = 4000\, e^{(1/2 \ln(2)\, t)}$$

```
> eqn4 := subs(t=3, eqn3);
```

$$eqn4 := y(3) = 4000\, e^{(3/2 \ln(2))}$$

```
> map(evalf, eqn4);
```

$$y(3) = 11313.70850$$

Exercise

(Exercise 4, page 471) Assuming exponential growth, find the number of bacteria present at 8:00 P.M. given that there are 7000 bacteria at 4:00 P.M. and the population triples every 8 hours.

Solution

Let $t = 0$ correspond to 4:00 P.M. Let $y(t)$ be the number of bacteria at time t.

```
> A := 7000;
```

$$A := 7000$$

```
> eqn1 := y(t) = A*exp(k*t);
```

$$eqn1 := y(t) = 7000\, \mathbf{e}^{(k\,t)}$$

```
> eqn2 := subs(t=8, rhs(eqn1)) = 3*subs(t=0, rhs(eqn1));
```

$$eqn2 := 7000\, \mathbf{e}^{(8\,k)} = 21000\, \mathbf{e}^{0}$$

```
> eqn3 := k = solve(eqn2, k);
```

$$eqn3 := k = \frac{1}{8} \ln(3)$$

```
> eqn4 := subs(eqn3, eqn1);
```

$$eqn4 := y(t) = 7000\, \mathbf{e}^{(1/8 \ln(3)\, t)}$$

```
> eqn5 := subs(t=4, eqn4);
```

$$eqn5 := y(4) = 7000\, \mathbf{e}^{(1/2 \ln(3))}$$

```
> map(evalf, eqn5);
```

$$y(4) = 12124.35566$$

Exercise

(Exercise 20, page 472) An E. coli population doubles every 20 minutes. What is the population size at the instant when the population is growing at the rate of 13,863 bacteria per minute.

Solution

Let $y(t)$ be the number of E. coli at time t.

```
> tau := 20;
```

$$\tau := 20$$

```
> eqn1 := diff(y(t),t) = k*y(t);
```

$$eqn1 := \frac{d}{dt}\, y(t) = k\, y(t)$$

```
> eqn2 := k = ln(2)/tau;
```

$$eqn2 := k = \frac{1}{20}\ln(2)$$

```
> eqn3 := subs(eqn2, eqn1);
```

$$eqn3 := \frac{d}{dt}\, y(t) = \frac{1}{20}\ln(2)\, y(t)$$

```
> eqn4 := 13863 = rhs(eqn3);
```

$$eqn4 := 13863 = \frac{1}{20}\ln(2)\, y(t)$$

```
> eqn5 := y(t) = solve(eqn4, y(t));
```

$$eqn5 := y(t) = \frac{277260}{\ln(2)}$$

```
> map(evalf, eqn5);
```

$$y(t) = 400001.6271$$

Exercise (Exercises 34 and 58, pages 473 and 476)

John Graunt, a ``collector and classifier of facts'', was the inventor of modern scientific demography. By tabulating the births and deaths listed in the *Weekly Bills of Mortality for London,* Graunt determined that the population of London was then growing exponentially with a doubling time of 64 years. (Graunt realized that the Exponential Law that he proposed could not be carried indefinitely backwards for then the city would be filled with ``far more People, then are now in it''.)



Natural and *Political*

OBSERVATIONS

Mentioned in a following INDEX,

and made upon the

Bills of Mortality.

By *JOHN GRAUNT,*

Citizen of

LONDON.

With reference to the *Government, Religion, Trade, Growth, Ayre, Diseases,* and the several Changes of the said CITY.

—— *Non, me ut miretur Turba, laboro, Contentus paucis Lectoribus* ——

LONDON,

Printed by *Tho: Roycroft,* for *John Martin, James Allestry,* and *Tho: Dicas,* at the Sign of the *Bell* in St. *Paul's* Church-yard, MDCLXII.

Accepting that

i) Graunt's growth observations were valid throughout the years 1662-1664 and 1666-1700 inclusive,
ii) the population of London declined by 100,000 during the Great Plague year 1665, and
iii) the population of London at the end of the year 1700 was about 500,000,

a) Estimate the population of London at the start of 1662, and
b) Plot the population of London for the years 1662-1700

Solution
Let $P(t)$ be the population of London in year t with $t = 0$ corresponding to January 1 1662. Then

$t = 1$ at the *end* of 1662,
$t = 2$ at the *end* of 1663,
$t = 3$ at the *end* of 1664,

and so on. In general, at the *end* of year Y with $1662 \le Y$, the value of t is $Y - 1662 + 1$.

Thus, if $T_1 = 3$, $T_2 = 4$, and $T_3 = 39$, then $t = T_1$ at the end of 1664 (which is to say, the beginning of 1665) , $t = T_2$ at the end of 1665 (which is to say, the beginning of 1666, and $t = T_3$ at the end of 1700. Let us make these definitions in Maple now

```
> T[1] := 3:   #Population at the en d of 1664, beginning of 1665
> T[2] := 4:   #Population at the end of 1665, beginning of 1666
> T[3] := 39:  #Population at the end of 1700
```

We are to determine $P(0)$.

For $t \le T_1$ London's population satisfies

$$P(t) = P(0)\, 2^{\left(\frac{t}{64}\right)}$$

Thus, in terms of the unknown $P(0)$, London's population at the beginning of 1665 was

$$P(T_1) = P(0)\, 2^{\left(\frac{T_1}{64}\right)} \qquad \text{(Population at the end of 1664, beginning of 1665)} .$$

The population $P(4)$ at the end of 1665, that is, at the start of 1666, was therefore given by

$$P(T_2) = P(0)\, 2^{\left(\frac{T_1}{64}\right)} - 100000 \qquad \text{(Population at the end of 1665, beginning of 1666)} .$$

At this point, exponential growth resumed. In the $T_3 - T_2$, or 35, years until the end of the year 1700, the population $P(T_2)$ increased by a factor $2^{\left(\frac{35}{64}\right)}$. In other words,

$$P(T_3) = P(T_2)\, 2^{\left(\frac{T_3 - T_2}{64}\right)} . \qquad \text{(Population at the end of 1700)} .$$

Putting everything together, we have

$$P(T_3) = \left(P(0)\, 2^{\left(\frac{T_1}{64}\right)} - 100000\right) 2^{\left(\frac{T_3 - T_2}{64}\right)} .$$

This population was 500000 and so we have an equation for $P(0)$:

$$500000 = \left(P(0)\, 2^{\left(\frac{T_1}{64}\right)} - 100000\right) 2^{\left(\frac{T_3 - T_2}{64}\right)} .$$

```
> eqn1 := 500000 = (P(0)*2^(T[1]/64)-100000)*2^((T[3]-T[2])/64);
```

$$eqn := 500000 = (\mathrm{P}(0)\, 2^{(3/64)} - 100000)\, 2^{\left(\frac{35}{64}\right)}$$

```
> eqn2 := P(0) = solve(eqn, P(0));
```

$$eqn2 := \mathrm{P}(0) = 50000\left(5 + 2^{\left(\frac{35}{64}\right)}\right) 2^{\left(\frac{13}{32}\right)}$$

```
> map(evalf, eqn2);
```

$$\mathrm{P}(0) = 428112.2504$$

We graph $\mathrm{P}(t)$ as follows

```
> P := t -> if t <= T[1] then 428112*2^(t/64)
            elif t <= T[2] then
  428112*2^(T[1]/64)-(t-T[1])*100000
            else (428112*2^(3/64)-100000)*2^((t-T[2])/64)
            end if:
> plot( P, 0..39, thickness=2, numpoints=600, tickmarks=[4,4],
  view=[0..39 , 300000..500000]);
```



Copyright and Author Information

Worksheet Title: BlankKrantz-06_5a-R8.mws   A Maple Release 8 worksheet.

Author: Brian E. Blank
Date Created: 30 September 2006 (Maple R8)
Date Last Revised: 29 September 2007

This document may not be distributed by any medium,
including print, disk, and electronic transfer, without
prior written permission of the author.

For more information, please contact the author:

Department of Mathematics,
Washington University in St. Louis
St. Louis, MO   63130

Telephone: (314) 935-6763
e-mail: brian@math.wustl.edu

Copyright: © 2006-2007 Brian E. Blank, All Rights Reserved.